Modelling the Temperature Dependent Reverse Recovery Behaviour of Power Diodes

Michael D. Reid, Simon D. Round and Richard M. Duke
Department of Electrical and Electronic Engineering,
University of Canterbury,
Private Bag 4800,
Christchurch, New Zealand
Phone: +64-3-366-7001, Fax: +64-3-364-2761

E-Mail: s.round@elec.canterbury.ac.nz

Abstract

A power diode PSpice model is presented for cryogenic use. This model can accurately simulate the diode's reverse recovery behaviour over all temperatures from 77K to room temperature. It has been tested over the full range of temperatures and the match with experimental data is excellent.

Key words: power semiconductor diodes, cryogenic electronics, modelling, SPICE

1 Introduction

Cryogenic operation of semiconductor devices has been shown to give significant benefits in power electronics [1]. The characteristics of MOSFETs and IGBTs are known to show a greater power efficiency when operated at low temperatures, while diodes have until recently been considered less suitable for cryogenic use as they have a higher voltage drop at lower temperatures under no-load conditions. Research performed at the University of Canterbury has found that this disadvantage is in some instances offset by advantages such as lower voltage drop under high current conditions, increased conductivity, lower reverse leakages currents and, as discussed in this paper, faster switching speeds [2].

Experimental reverse recovery current is illustrated in Figure 1 for operating temperatures from 300K to 77K. The peak reverse current can, for high temperatures, be comparable to or even greater than the on-state forward current flow. During the switching time while this reverse current is flowing, the diode is supporting a large reverse voltage, resulting in a substantial power loss. The reverse recovery current and hence the power loss are greatly reduced by operating at cryogenic temperatures.

To fully evaluate the benefits of diode use at temperatures between 77K and room temperature, simulation tools are essential. As no cryogenic model is available for this purpose, a new diode model has been developed for the widely known simulation package PSpice. This model adds functionality to a temperature-dependent static model previously developed by the authors [2].



Figure 1. Reverse recovery of the DTV32 diode at various temperatures.

2 Review of Diode Modelling Techniques

The default diode model in PSpice assumes that the dynamic behaviour can be described by a capacitance, leading to the inaccurate simulation shown in Figure 2. In the presence of a stray circuit inductance, the PSpice model leads to oscillation after the diode turn-off that does not match the experimental results. Many subcircuit models have been proposed which attempt to overcome this deficiency, and a thorough overview is presented by Tan and Tseng [3].

Two types of model are available: analytic and numerical/hybrid. The latter method was rejected for this work, as the resulting models are often too complex and processor-intensive for use in PSpice. Of the analytic modelling methods, the lumped-charge method was selected. This gives a number of advantages, including an insight into the internal workings of the device, and allowing comparisons



Figure 2. Reverse recovery of the PSpice built-in diode. Experimental data was taken at 300K and the diode model parameters were fitted to give the closest approximation to this data.

with previously known temperature dependencies, such as that of the ambipolar diffusion constant.

The best known lumped-charge work has been done by Lauritzen and Ma [4]. The model they proposed has foundations in semiconductor theory, and was found to be simple to implement and convergently robust. Using their approach, a model has been derived which can accurately model the reverse recovery of a power diode over the temperature range 77 to 300K.

In the characteristics shown in Figure 1, a tail current is observed at temperatures above 260K. This tail current decays slowly to zero over more than a microsecond, following the initial fast drop in reverse current which takes less than 200 nanoseconds. Many diodes show such behaviour, and slower diodes show a tail current at all temperatures. This behaviour is not modelled by any of the analytical methods covered by Tan and Tseng [3], nor is it modelled by many numerical methods. The analytical modelling methods all assume a single time constant for the current decay, and do not allow modelling of the fast initial current snap-off.

3 Improved Dynamic Model

The Lauritzen and Ma model contains no temperature dependency. The temperature dependent dynamic parameters must be identified in order to implement the behaviour shown in Figure 1. The critical parameters are found to be the lifetime, τ, and the drift region transit time, T_M.

A basic extraction method for these parameters is given by Lauritzen and Ma. Their method is not sufficiently accurate for use when noise exists in the



Figure 3. Observed reverse-recovery waveform for power diodes, with important features labelled.

data, and so an enhanced extraction procedure is presented here. The diode used is a DTV32-1200A diode, with maximum ratings of 1200V and 6A, and a nominal reverse recovery time of 600ns. This diode was chosen because its characteristics over the temperature range of interest are representative of all power diodes investigated.

3.1 Parameter Extraction

Turn-off refers to the period of the reverse recovery waveform when diode current is decreasing toward the reverse current peak. The snap-off period occurs while the current is rising rapidly following the reverse peak. These features are illustrated in Figure 3.

The DTV32 reverse recovery characteristics shown in Figure 1 display constant turn-off and snap-off slopes. This is typical of all five diodes that were investigated. The turn-off slope a must be constant with temperature, as it is the ratio of two temperature-independent quantities,

$$a = \frac{V_0}{L_s} \qquad (1)$$

where V_0 is the voltage acting reverse-bias the diode, and L_s is the series circuit inductance. Lauritzen and Ma introduce some further quantities. These are T_1, the length of the turn-off period, I_{RM}, the maximum reverse current, and τ_{rr}, the time constant of the decay tail. Equation (2) relates τ_{rr} to the two dynamic parameters, lifetime and transit time [4].

$$\frac{1}{\tau_{rr}} = \frac{1}{\tau} + \frac{1}{T_M} \qquad (2)$$

The values of T_1, I_{RM} and τ_{rr} cannot be measured accurately in the presence of noise. Thus the extraction method proposed by Lauritzen and Ma is not possible. Instead, the total reverse charge, Q_{rr}, is calculated. This quantity is found by integration of the current (3), allowing noise to be eliminated.

$$Q_{rr} = \int_{i(t)<0} i(t)\,dt \qquad (3)$$

Matching Q_{rr} between experiment and the model yields (4), while matching the reverse charge allows the snap-off slope b to be related to the decay time constant (5).

$$I_{RM} = \sqrt{\frac{Q_{rr}}{\frac{1}{2a} + \frac{1}{2b}}} \qquad (4)$$

$$b = \frac{1}{2\tau_{rr}} \qquad (5)$$

The parameter extraction routine is completed by taking (6), which is given by Lauritzen and Ma, and expanding it into the form given in (7). The values of τ and T_M are found by simultaneous solution of (2), (4), (5) and (7).

$$I_{RM} = a\,(\tau - \tau_{rr})\left[1 - \exp\left(-\frac{T_1}{\tau}\right)\right] \qquad (6)$$

$$\tau^2\left(e^{-\frac{I_{RM}+I_F}{a\tau}} - 1\right) = -\frac{I_{RM}}{a}(\tau + T_M) \qquad (7)$$

3.2 Temperature Dependencies of DTV32-1200A Parameters

By averaging over all temperatures, the experimental values of a and b were found to be $a = 12.06A/\mu s$ and $b = 14.59A/\mu s$. The extracted values of Q_{rr} were found by integrating from the point where the current first goes negative to the point where it next goes positive. This ending point will be close to the point where the current falls to zero — the current goes positive due to noise. To eliminate noise on the negative-going edge, a restriction is placed such that current must reach a negative value of 5% of the original forward current before integration can commence.

The results are shown in Figure 4. As can be seen, this dependence is highly piecewise linear, with a corner temperature of 260K. A linear fit was made to this curve below 260K, and extrapolated to 300K. It was assumed that the corner was due to the appearance of the exponential current tailing effect, which significantly increases the reverse charge due to a long decay time. This additional component of the reverse current was therefore ignored, as this single time constant model is not able to simulate multi-part decay waveforms.

While this appears to be the correct explanation of the two-part linear behaviour of Q_{rr}, the verification will show that the linear extrapolation is not exact up to 300K.



Figure 4. Extracted reverse recovery charge for the DTV32 diode.



Figure 5. Temperature dependence of the dynamic diode parameters; Lifetime τ, and Transit time T_M. (dotted) Extracted (solid) Fitted.

Using this linear fit to Q_{rr}, the values of a and b found above, and the experimentally specified forward current value of $2.5A$, the values of τ and T_M were extracted, and are graphed in Figure 5. The most accurate fitting formula for both quantities was found to be of the form

$$\tau \sim (\alpha T + \beta)^{\frac{1}{\gamma}}$$

These fits are also shown in Figure 5. This completes the specification of the dynamic component of a DTV32-1200A cryogenic diode model. A listing of the completed model is given in Appendix A.

4 Verification

The test circuit used for the PSpice simulation of this model is shown in Figure 6. In practice, the ideal



Figure 6. The test circuit used for verification.

switch was approximated by a $MTP3055$ MOSFET and the constant current source by an $8mH$ inductor in series with 2.2Ω resistance. The stray inductance L_s was measured as $1.67\mu H$.

Dynamic characteristics match well between this model and experiment, as shown for a selection of temperature points in Figure 7. Using an exponential decay to simulate an observed linear snap-off gives reasonable accuracy for temperatures up to 260K. Above this temperature, an exponential decay appears in the experimental characteristic which has a longer time constant than the modelled behaviour. Moreover, as seen in the 300K characteristic, the derived model underestimates the width of the reverse current peak, leading to a value for the reverse recovery charge that is too low. This indicates that the extrapolation in Figure 4 is not exact up to 300K. For a more accurate simulation, a model is needed which explains the observed two-part decay waveform.

The extraction method proposed here is straightforward and may be used for any diode which has little or no appearance of the two-part decay waveform over the temperature range of interest.

5 Future Work

The experimental reverse recovery waveforms do not show an exponential decay as assumed by all analytic diode models as reviewed by Tan and Tseng [3]. A model which accounts for this effect must include a transit time capacitance as modelled by the built-in PSpice diode model. The model must furthermore recognise the fundamental differences in nature between ambipolar stored charge, as contained in the drift region of a power semiconductor device, and electrostatic charge as stored by a capacitor. Such a model has been developed by the authors and



Figure 7. Temperature dependence of the dynamic diode characteristics (solid) Experimental, and (dotted) New model.

work is currently in progress to expand this into a temperature-dependent model. It is expected that the results will be presented in a future paper.

6 Conclusions

A new model has been created in PSpice that improves the accuracy of reverse recovery modelling at temperatures in the cryogenic range from 77 to 300K. The method presented may be used to create a model for any diode for which experimental reverse recovery data exists.

In order to develop this temperature-dependent model, a new parameter extraction routine was devised and applied to the data from a DTV32-1200A device. The new extraction routine uses charge and slope values from the electrical data, rather than peak values and exponential time constants. This allows the new method to be significantly more robust than previous methods in the presence of noise.

Like most analytic diode models, a single decay time constant is assumed. This is contradictory to the experimental data which shows a sharp snap-off before the exponential decay commences. A fully accurate model would thus need to account for this effect, and a discussion is given on the possibility of future work in this area.

Acknowledgment

Thanks go to Prof. W. P. Robbins, University of Minnesota, for providing the test circuit and to A. P. R. Taylor for the experimental data.

References

[1] R. Singh and B. J. Baliga, *Cryogenic operation of silicon power devices,* Kluwer Academic Publishers, 1998.

[2] A. P. R. Taylor, M. D. Reid, R. M. Duke and S. D. Round, "Experimental characteristics and an enhanced PSpice model for power diodes from 77K to 300K," Proc. AUPEC/EECON '99, pp. 252-257, Sep 1999.

[3] C. M. Tan and K. J. Tseng, "Using power diode models for circuit simulations – a comprehensive review," IEEE Trans. on Industrial Applications, vol. IA-46, no. 3, pp. 637-645, 1999.

[4] P. O. Lauritzen and C. L. Ma, "A simple diode model with reverse recovery," IEEE Trans. on Power Electronics, vol. PE-6, no. 2, pp. 188-191, 1991.

A Model Listing

```
.SUBCKT DTV32-1200A 1 9 PARAMS:
+ CAP=15PF
+ GDE=0.5
+ FBCOEFF=0.5
+ PHI=1

* Cryogenic power diode model.

* Temperature information is
* passed via the TEMP and VT
* parameters.

.FUNC H(X)={0.5*(X+ABS(X))}
* Heaviside unit-ramp function.

.PARAM RS={.08534038
+ +423.07661e-9*(TEMP+48.435)^2
+ +246.5547e-6*H(-175.15-TEMP)}
.PARAM TV0=1.4326
.PARAM TV1=-1.2214E-3
.PARAM IM0=2

.PARAM TAU0=-6.8423E-11
.PARAM TAU1=1.12E-12
.PARAM TAU2=1.5
.PARAM TM0=-1.2657E-16
.PARAM TM1=2.0379e-18
.PARAM TM2=2.2
* Implicitly requires T>63K

.PARAM VE0={(TV0+(TEMP+273.15)*TV1-RS*IM0)/2}
.PARAM QE0={IM0*(TM+TAU)}
.PARAM TAU={(TAU1*(TEMP+273.15)+TAU0)^(1/TAU2)}
.PARAM TM={(TM0+(TEMP+273.15)*TM1)^(1/TM2)}
.PARAM RATIO={TAU*TM/(TAU+TM)}
.PARAM ISTAU={LIMIT(QE0/(EXP(VE0/VT)-1),-1E-25,1E4)}
* The product Is*tau.

.PARAM XA={2E-28*EXP(-16.22*(TEMP+273.15)/300)}
.PARAM XB={5.8+1.2*(TEMP+273.15)/300}
.PARAM W=44E-4
.PARAM VBRK={W^((XB-1)/XB)*(2/XA)^(1/XB)*(XB/(XB+1))}
* Breakdown variables, based on the
* working of Singh and Baliga [1].

GD 1 2 VALUE={(V(5)-V(6))/TM-IM0}
DCJ 1 2 DCAP
.MODEL DCAP D(IS=1E-25, RS=0, TT=0, CJO={CAP}, M={GDE}
+ FC={FBCOEFF}, VJ={PHI}, IBV=2, BV={VBRK})

EE 5 0 VALUE={LIMIT((QE0+ISTAU)*EXP(0.5*V(1,2)/VT)
+ -ISTAU,-1,1)}
RE 5 0 1E6
EM 6 0 VALUE={LIMIT((V(5)/TM-I(VSENSE1))*RATIO,-1,1)}
RM 6 0 1E6

EDM 7 0 VALUE={V(6)}
VSENSE1 7 8 DC 0
CDM 8 0 1
RDM 8 0 1E9

V0 2 4 {2*VE0}
IM0 1 4 {IM0}
RS1 4 9 {RS}
* This model uses differential static modelling,
* as discussed by Taylor, Reid, Duke and Round [2].

.ENDS DTV32-1200A
```